CONSTRUCTIONS OF GORENSTEIN RINGS
GUNTRAM HAINKE AND ALMAR KAID

1. Introduction
This is a summary of the fourth tutorial handed out at the CoCoA
summer school 2005. We discuss two well known methods of construction of Gorenstein rings using Liaison Theory and so called BuchsbaumRim modules. In particular, we provide the necessary CoCoA (version
4.6) codes for implementation.
Let K be a field and P := K[x0 , . . . , xn ] the polynomial ring equipped
with the standard grading.
Definition 1.1. Let I ⊂ P be a homogeneous ideal. The graded P algebra A := P/I is called Gorenstein if the following holds:
(1) A is Cohen-Macaulay.
(2) The last module in the minimal graded free resolution of A is
a free module of rank one.
The K-vector space soc A := ((0) :A m) = {x ∈ A : m · x = 0} is
called the socle of A, where m := (x0 , . . . , xn ).
Proposition 1.2. Suppose A is Artinian. Then the following conditions are equivalent:
(1) A is Gorenstein.
(2) soc A is a K-vector space of dimension one.
Proof. Since A is Artinian we have dim A = 0. Hence A is CohenMacaulay. Note that soc A ∼
= HomP (K, A) = Ext0P (K, A) and by
[1, Exercise 3.3.26], we have dimK Ext0P (K, A) = dimK TorPn+1 (K, A).
Since A is Artinian the Auslander-Buchsbaum formula (cf. [1, Theorem 1.3.3]) yields that dimK TorPn+1 (K, A) equals the rank of the last
free module in the graded free resolution of A and we are done.
¤
Now we want to develop a CoCoA function which checks whether
the algebra A = P/I is Gorenstein. To do this we use the following
two subfunctions:
1

Define ProjDim(I)
B:=BettiMatrix(I);
Return(Len(B[1]));
EndDefine;
Define IsCohenMacaulay(I)
Pd:=ProjDim(I);
Codim:=NumIndets()-Dim(CurrentRing()/I);
Return Pd=Codim;
EndDefine;
The function ProjDim(I) computes the projective dimension pd A
of A = P/I. Next the function IsCohenMacaulay(I) checks whether A
is Cohen-Macaulay and returns the corresponding Boolean value. Note
here that A is Cohen-Macaulay if and only if pd A = codim I, where
codim I = min{ht(p) : I ⊆ p minimal}.
By using the functions ProjDim(I) and IsCohenMacaulay(I) we
can define the function IsGorenstein(I) which checks whether A =
P/I is Gorenstein and returns the corresponding Boolean value:
Define IsGorenstein(I)
B:=BettiMatrix(I);
C:=Len(B);
Return IsCohenMacaulay(I) AND (B[C][1])=1;
EndDefine;

2. Using Liaison Theory
In this section we use Liaison Theory to construct graded Gorenstein
algebras. Our main tool is the following theorem (cf. also [3, Example
6.5(3)]).
Theorem 2.1. Let P/I1 and P/I2 be two graded Cohen-Macaulay rings
of the same dimension d and suppose that P/(I1 ∩ I2 ) is Gorenstein.
Moreover, suppose that I1 and I2 have no prime component in common.
Then P/(I1 + I2 ) is a graded Gorenstein algebra of dimension d − 1.
Proof. Let c := n + 1 − d denote the codimension of I1 and I2 . The
minimal free resolutions of I1 ∩ I2 , I1 and I2 combined with the exact
sequence
0 −→ I1 ∩ I2 −→ I1 ⊕ I2 −→ I1 + I2 −→ 0

lead to the commutative diagram
0

0

²

P (−k)
²

Fc−1
²

..
.
²

F2
²

0

²

Gc ⊕ Hc
²

Gc−1 ⊕ Hc−1
²

..
.
²

G2 ⊕ H2
²

F1

G1 ⊕ H1

²
/ I1 ∩ I2

²
/ I1 ⊕ I2

²

0

/ I1 + I2

/0

²

0.

If we apply the mapping cone to the resolutions of I1 ∩ I2 and I1 ⊕ I2
we get the (not necessarily minimal) free resolution
0 −→ P (−k) −→ Fc−1 ⊕ (Gc ⊕ Hc ) −→ · · · −→ F1 ⊕ (G2 ⊕ H2 )
−→ G1 ⊕ H1 −→ I1 + I2 −→ 0
for the ideal I1 + I2 . Hence pd P/(I1 + I2 ) ≤ c + 1. Since I1 and
I2 have no common component we have codim P/(I1 + I2 ) ≥ c +
1 and dim P/(I1 + I2 ) ≤ n + 1 − (c + 1) = n − c. In particular,
depth P/(I1 + I2 ) ≤ dim P/(I1 + I2 ) ≤ n − c (cf. [1, Proposition
1.2.12]). The Auslander-Buchsbaum formula (cf. [1, Theorem 1.3.3])
gives pd P/(I1 +I2 ) ≥ n+1−(n−c) = c+1. Hence pd P/(I1 +I2 ) = c+1.
Again by the Auslander-Buchsbaum formula we get depth P/(I1 +I2 ) =
n − c. So the estimate above yields depth P/(I1 + I2 ) = n − c =
dim P/(I1 + I2 ), i.e. P/(I1 + I2 ) is Cohen-Macaulay. Since P/(I1 + I2 )
has type one, it is also Gorenstein.
¤
Definition 2.2. Let V ⊂ Pn be a subscheme with homogeneous coordinate ring R := P/IV (we assume that IV is saturated).
(1) V is arithmetically Cohen-Macaulay if R is Cohen-Macaulay.
(2) V is arithmetically Gorenstein if R is Gorenstein.

Theorem 2.1 has the following geometric formulation: Let V1 , V2 ⊂
Pn be arithmetically Cohen-Macaulay subschemes of codimension c
with no common component. Assume that X := V1 ∪ V2 is arithmetically Gorenstein. Then V1 ∩ V2 is arithmetically Gorenstein of
codimension c + 1.
For instance, consider two arithmetically Cohen-Macaulay curves
which are geometrically linked, i.e. they have no common component
and their union is a complete intersection. In this case Theorem 2.1
asserts that the intersection of these curves is an arithmetically Gorenstein set of points. To illustrate this we provide the following example.
Example 2.3. We start with the twisted cubic curve C in P3 , i.e. the
curve defined by the ideal IC which is generated by the 2 × 2 minors of
the matrix
µ
¶
x0 x1 x2
.
x1 x2 x3
To realise this situation in CoCoA we use the command
Use P::=Q[x[0..3]];
to work in the polynomial ring in four variables. We get the ideal IC
in the following way:
Matrix:=Mat([[x[0],x[1],x[2]],[x[1],x[2],x[3]]]);
IC:=Ideal(Minors(2,Matrix));
By using
IsCohenMacaulay(IC);
we see that C is an arithmetically Cohen-Macaulay curve. Furthermore
we use the following function:
Define IsCompleteIntersection(I)
Codim:=NumIndets()-Dim(CurrentRing()/I);
Return Len(MinGens(I))=Codim;
EndDefine;
It checks whether P/I is a complete intersection in terms of the length
of a minimal system of generators.
If we apply IsCompleteIntersection(I) to the ideal IC , we see
that the twisted cubic curve is not a complete intersection. Now we
write another very useful function:
Define GenRegSeq(I,L)
M:=Ideal(Indets());
Seq:=[];
N:=Ideal(0);

Foreach T In L Do
J:=Intersection(M^T,I);
D:=Deg(Head(MinGens(J)));
If D<>T Then
PrintLn"No regular sequence of this type possible";
Return;
EndIf;
L2:=[F In MinGens(J) | Deg(F)=D];
S:=Sum([Rand(-10,10)*F | F In L2]);
If N:Ideal(S)<>N Then
PrintLn"No regular sequence of this type possible";
Return;
EndIf;
Append(Seq,S);
N:=Ideal(Seq);
EndForeach;
Return Seq;
EndDefine;
The function GenRegSeq(I,L) takes an ideal I and a list of degrees
L and computes a regular sequence of homogeneous polynomials in I
whose degrees are given by L. If this is not possible it conveys this
information.
Now we get by
ID:=Ideal(GenRegSeq(IC,[3,3]));
the ideal ID of a complete intersection curve D of type (3, 3) containing
C. Moreover, we get the vanishing ideal IE of the residual curve E :=
D\C by computing the colon ideal
IE:=ID:IC;
We use the command
Hilbert(P/IE);
to see that the Hilbert polynomial of E is HPE (t) = HPP/IE (t) = 6t−2
for t ≥ 1. Hence E is a (arithmetic) genus three curve of degree 6.
Further, with the help of
IsCohenMacaulay(IE);
we notice that E is arithmetically Cohen-Macaulay.
Now set X := C ∩ E. By Theorem 2.1 the scheme X is an arithmetically Gorenstein set of points. We compute its vanishing ideal IX
by

IX:=IE+IC;
The command
Hilbert(P/IX);
yields the constant Hilbert polynomial HPX (t) = HPP/IX (t) = 8 for
t ≥ 3 of X. Hence X is a zero-dimensional scheme of degree 8. In
particular, if we use our two CoCoA functions IsGorenstein(I) and
IsCompleteIntersection(I) applied to the ideal IX , we verify that X
is arithmetically Gorenstein but not a complete intersection. Thus we
have constructed a “new” 1-dimensional graded Gorenstein ring P/IX .
Example 2.4. We will give an analog example to the previous one and
construct the ideal X of 11 points on the twisted cubic curve C defined
above. Here, we utililize a complete intersection of type (3, 4). So we
get in the same way the ideal IX via the commands:
ID:=Ideal(GenRegSeq(IC,[3,4]));
IE:=ID:IC;
IX:=IE+IC;
To check whether X is a reduced scheme, we query
IX=Radical(IX);
and get IX = rad(IX ). If we compute the Hilbert polynomial of X with
Hilbert(P/IX);
we get indeed HPX (t) = HPP/IX (t) = 11 for t ≥ 4. Hence (at least
after extension of the base field) X is a set of 11 points in Pn . We
verify Theorem 2.1 with
IsGorenstein(IX);
IsCompleteIntersection(IX);
and notice that X is arithmetically Gorenstein but not a complete intersection.
3. Using Buchsbaum-Rim modules
We still work in the polynomial ring P = K[x0 , . . . , xn ] over a field
K. First we recall the notion of a Buchsbaum-Rim module.
L

L

t
Definition 3.1. Let F := t+r
j=1 P (−bj ) be two
i=1 P (−ai ) and G :=
graded free P -modules, where t, r ∈ N and a1 , . . . , at+r , b1 , . . . , bt ∈ Z.
Furthermore, let Φ : F → G be a homogeneous map. The map Φ
is given by a t × (t + r) matrix A whose entries are homogeneous
polynomials. The module B := ker Φ is called the Buchsbaum-Rim

module associated to Φ (or to A) if the ideal of maximal minors of A
has the expected codimension r + 1.
e on Pn associated to B is also called the
Remark 3.2. The sheaf B
Buchsbaum-Rim sheaf associated to Φ (or to A). In the following it
will be not necessary to use the sheaf framework.

We recall that the top dimensional part I top of an ideal I is the intersection of its primary components of maximal dimension. The following theorem (cf. [4, Theorem 4.7]) provides a method of constructing
Gorenstein algebras via Buchsbaum-Rim modules.
Theorem
3.3. Let B be L
the Buchsbaum-Rim module associated to a
Lt+r
map Φ : i=1 P (−ai ) → tj=1 P (−bj ), s ∈ B a generic homogeneous
element and let I ⊆ P = K[x0 , . . . , xn ] be the ideal generated by the
components of s. If r < n is odd, then the top dimensional part I top
yields a Gorenstein ring P/I top .
Now we give an example of an application of Theorem 2.1.
Example 3.4. Let C ⊂ P4 be the curve defined by the vanishing of
the 2 × 2 minors of the matrix
µ

¶

x0 x1 x2 x3
.
x1 x2 x3 x4

In the following we apply Theorem 3.3 to find an arithmetically Gorenstein curve D of small degree containing it. We use the command
Use P::=Q[x[0..4]];
to work with CoCoA in the polynomial ring in five variables over the
rationals. Now we want to compute the Buchsbaum-Rim module B
associated to the matrix A := (x0 , x1 , x2 , x3 ). To do this, we compute
the syzygy module for the variables x0 , x1 , x2 , x3 :
B:=Syz(x[0]..x[3]);
We notice that B is contained in the free P -module F := P (−1)4 .
Further, we get the vanishing ideal IC of the curve C by
IC:=Ideal(Minors(2,Mat([x[0]..x[3],x[1]..x[4]])));
Since we want D to contain C, we use an element s ∈ Bd , d ∈ Z,
whose components are contained in IC . Thus we have to compute
B 0 := B ∩ IC F . Firstly, we compute the module IC F in the following
way:

GenIC:= Gens(IC);
ICF:=[];
For I:=1 To 4 Do
Foreach X In GenIC Do
Append(ICF,X * E_(I,4));
EndForeach;
EndFor;
ICF:=Module(ICF);
Hence we get B 0 by
Bprime:=Intersection(B,ICF);
Now we construct a generic element s ∈ B 0 of degree 2. To do this, we
firstly compute all generators of degree 2 of B 0 by
Bprime_2:=[X In Gens(Bprime)|Deg(X)=2];
To get a generic (in the randomized sense) element s ∈ B20 we use the
command Rand(-10,10) which produces a random integer between
−10 and 10 (of course one can use other parameter or use Rand() to
produce an “arbitrary random integer”):
S:=Sum([Rand(-10,10)*X | X In Bprime_2]);
We denote by I the ideal generated by the components of s and realize
this with
I:=Ideal([X | X In List(S)]);
Let D be the projective scheme defined by I(= ID ). To describe D,
we use the function EquiIsoDec(I) which computes a list of unmixed
ideals I1 , . . . , Ik such that rad(I) = rad(I1 ) ∩ · · · ∩ rad(Ik ). So the
computation
L:=EquiIsoDec(I);
yields ideals L[2], L[3], L[5], L[6] (L[1] and L[4] equal the unit ideal and
are therefore obviously redundant). We use the function Radical(I) to
check that all these ideals are radical ideals. Next, we use the function
Intersection(I,J) to conclude the inclusions L[2] ⊂ L[3], L[2] ⊂ L[5]
and L[2] ⊂ L[6]. Hence rad(I) = L[2]. Next we check by
Hilbert(P/I);
Hilbert(P/L[2]);
that P/I has Hilbert polynomial HPP/I (t) = 5t + 1 and P/L[2] has
Hilbert polynomial HPP/L[2] (t) = 5t. Therefore they differ by a “point”
in the sense of the exact sequence
0 −→ L[2]/I −→ P/I −→ P/L[2] −→ 0.

top
Finally, we want to compute the top dimensional part of D, i.e. ID
.
The following lemma provides an effective method of computing the top
dimensional part of an ideal.

Lemma 3.5. Let I ⊆ P be a saturated ideal and J ⊆ I be an ideal
generated by a regular sequence whose length is the codimension of I.
Then
J :P (J :P I) = I top .
Proof. For a detailed proof see [2, Proposition 5.2.3(d), Remark 5.1.5
and the proof on p. 123]. If X denotes the arithmetically Gorenstein scheme defined by J, then basically one shows that the projective
schemes defined by the ideals I top and J :P I are algebraically CI-linked
via X.
¤
Now we apply Lemma 3.5 to our example. So we compute an ideal
J generated by a regular sequence of length codim I = 3 in I:
J:=Ideal(GenRegSeq(I,[2,2,2]));
According to Lemma 3.5 we get the top dimensional part of I by
Itop:=J:(J:I);
To describe Dtop := Proj(P/I top ) we use
Hilbert(P/Itop);
and get HPDtop (t) = HPP/I top (t) = 5t for t ≥ 1, i.e. Dtop has arithmetic
genus pa (Dtop ) = 1 and degree 5. By Theorem 3.3 the curve Dtop is
arithmetically Gorenstein. We can describe Dtop a bit more. We check
via
Intersection(IC,Itop)=Itop;
that IDtop ⊂ IC , i.e. the curve C is contained in Dtop . Further, we
compute the residual curve E := Dtop \C with
IE:=Itop:IC;
and its Hilbert polynomial by
Hilbert(P/IE);
We get HPE (t) = HPP/IE (t) = t + 1 for t ≥ 0. Hence the top dimensional part Dtop of D is the union of C with a line.

